

# Wire Heat Transfer for CSO Dewars

F. Rice

5/28/2010

## Part II: Optimal magnet wire design

---

Execute the Part I notebook to get the functions defined there

```
In[1]:= Needs["Units`"];

In[2]:= If[Head[TC] != InterpolatingFunction,
  part1 = NotebookOpen[ToFileName[NotebookDirectory[], "Wire Heat Xfer I.nb"]];
  SelectionMove[part1, All, Notebook];
  SelectionEvaluateCreateCell[part1]
]
```

---

## Optimizing the dimensions of a wire carrying current

Source: [http://www.submm.caltech.edu/cso/receivers/thermal/opt\\_wiredimensions.pdf](http://www.submm.caltech.edu/cso/receivers/thermal/opt_wiredimensions.pdf)

■ Since  $\rho$  is essentially constant, it can be factored out of the optimizing integral

The optimal ratio of length to area for a wire carrying current  $I$ , for nearly constant resistivity  $\rho$ :

$$\left(\frac{L}{A}\right)_{\text{Optimum}} = \frac{1}{I} \int_{T_L}^{T_H} \frac{\kappa}{\sqrt{2 \int_{T_L}^{T_H} \rho \kappa dT'}} dT \approx \frac{1}{I \sqrt{\rho}} \int_{T_L}^{T_H} \frac{\kappa}{\sqrt{2 \int_{T_L}^{T_H} \kappa dT'}} dT$$

The last expression above can be integrated analytically, so for nearly constant resistivity  $\rho$ :

$$\left(\frac{L}{A}\right)_{\text{Optimum}} \approx \frac{1}{I \sqrt{\rho}} \sqrt{2 \int_{T_L}^{T_H} \kappa dT} \Rightarrow \boxed{(\text{Ohmic dissipation}) = 2 \times (\text{Thermal heat transfer})}$$

So optimum  $L/A$  should result in the wire's Ohmic dissipation being equal to twice the heat transfer in the wire (when current = 0). An optimum wire bundle will transfer all Ohmic heating to the cold stage, but no net heat transfers from the hot stage.

Given the optimal  $L/A$  ratio for current  $I$ , the heat transfer to the cold surface is:

$$\boxed{(Q_L)_{\text{optimum}} = I \sqrt{2 \int_{T_L}^{T_H} \rho \kappa dT} = I V_{\text{optimum}}, \quad V_{\text{optimum}} \equiv \sqrt{2 \int_{T_L}^{T_H} \rho \kappa dT}}$$

There is an optimal voltage drop across the wire,  $V_{\text{optimum}}$ , for minimum heat transfer which depends on the material properties and the end-point temperatures, but is independent of the design current  $I$ .

### ■ A function to calculate the optimum L/A ratio for Manganin wire carrying a specified current

The units factor will handle  $I$  in mA and  $\rho$  in  $\mu\Omega$  cm, along with the units of the integral above, so that the RHS of the equation is in  $\text{Meter}^{-1}$ :

```
In[31]:= Clear[units];
units =
  Sqrt[Meter^2 Convert[1.0 Watt / Meter / (Milli Amp)^2 / (Micro Ohm Centi Meter), Meter^-2]] / Meter

Out[32]= 1. x 10^7
         Meter
```

This function calculates the optimum L/A (in  $\text{Meter}^{-1}$ ) for a given current (and optional temperature limits):

```
In[33]:= Clear[LbyAopt];
LbyAopt[current_?NumberQ(* milliamps *), Tlow_?NumberQ(* Kelvin *),
  Thigh_?NumberQ(* Kelvin *)] := units Sqrt[2 TI[Tlow, Thigh]] / (current Sqrt[rho])

LbyAopt[current_?NumberQ(* milliamps *)] := LbyAopt[current, 4.0, 15.0]
```

Now use the above function to calculate the optimum length for a bundle of wires:

### ■ A function to calculate the optimum wire length given:

- the number of wires
  - the diameter of each wire (mils)
  - the total current (mA)
- for Manganin wire connecting 4K to 15K

```
In[36]:= Clear[OptLen];
OptLen[n_, dia_(* mils *), current_(* milliamps *)] :=
  Convert[n area[dia] Meter^2 LbyAopt[current], Inch]
```

### ■ The optimum length for a 4-wire bundle of 5 mil Phosphor Bronze wires, carrying a total of 10 mA:

```
In[38]:= SetMaterial["Phosphor Bronze"]

In[39]:= OptLen[4(* wires *), 5.(* mils *), 10.(* milliamps *)]

Out[39]= 6.61295 Inch
```

- A function to calculate the optimum number of wires to carry a current over a specified length (for wire connecting 4K to 15K), given:  
the length (inches)  
the diameter of each wire (mils)  
the total current (mA)

```
In[40]:= Clear[OptN];
OptN[Len_ (* inches *), dia_ (* mils *), current_ (* milliamps *)] :=
  Len Inch / OptLen[1.0, dia, current]
```

- The optimum number of wires for a 4 inch length of 5 mil wires, carrying a total of 10 mA:

```
In[42]:= OptN[4. (* inches *), 5. (* mils *), 10. (* milliamps *)]
```

```
Out[42]= 2.41949
```

---

## Heat transfer of an optimal wire bundle carrying current

- A function to calculate the heat transfer to the 4K stage given an optimum length/area wire bundle

An optimum L/A bundle will have a voltage drop determined by the temperatures, wire resistivity and thermal conductivity. For Manganin, this is ~4mV for a 15K - 4K connection; for Phosphor Bronze it is ~3mV, better than Manganin by about 25%:

```
In[43]:= Clear[OptVoltage];
OptVoltage[Tlow_ (* Kelvin *), Thigh_ (* Kelvin *)] :=
  Volt  $\sqrt{\text{Convert}[2. \text{ rho } (\text{Micro Ohm Centi Meter}) \text{ TI}[Tlow, Thigh] (\text{Watt / Meter}), \text{Volt}^2] / \text{Volt}^2}$ 
OptVoltage[] := OptVoltage[4., 15.]
```

```
In[46]:= Clear[OptHeat];
OptHeat[current_ (* milliamps *)] :=
  Convert[ current ( Milli Amp) OptVoltage[], Micro Watt]
OptHeat[current_ (* milliamps *), Tlow_ (* Kelvin *), Thigh_ (* Kelvin *)] :=
  Convert[ current ( Milli Amp) OptVoltage[Tlow, Thigh], Micro Watt]
```

```
In[49]:= SetMaterial["Manganin"];
Print["Manganin: ", OptVoltage[], "; heat for 10mA: ", OptHeat[10]]
SetMaterial["Phosphor Bronze"];
Print["Phosphor Bronze: ", OptVoltage[], "; heat for 10mA: ", OptHeat[10]]
```

```
Manganin: 0.00405375 Volt; heat for 10mA: 40.5375 Micro Watt
```

```
Phosphor Bronze: 0.00291712 Volt; heat for 10mA: 29.1712 Micro Watt
```

- Summary of optimal results for a 4 inch bundle carrying 10 mA of magnet current

- Magnet current on :

```
In[53]:= OptHeat[10 (* milliamps *)]
```

```
Out[53]= 29.1712 Micro Watt
```

Magnet current off (the function Heat[ ] is from the Part I notebook) :

```
In[54]:= Heat[OptN[4.( * inches *), 5.( * mils *), 10.( * milliamps *)],
          4.( * inches *), 5.( * mils *)]
```

```
Out[54]= 14.5856 Micro Watt
```

## Heat transfer for non-optimum L/A

The optimum L/A for a wire bundle is inversely proportional to the current, so if a bundle is optimally designed for a specific current, then it will be too thick for smaller currents and too thin for larger currents. In this section we calculate the heat transfer to the cold stage for these non-optimal currents.

The integral equation for the heat transfer in the non-optimal L/A case is (the second expression is valid for nearly constant resistivity  $\rho$ ):

$$\frac{L}{A} = \int_{T_L}^{T_H} \frac{\kappa}{\sqrt{Q_H^2 + 2 I^2 \int_T^{T_H} \rho \kappa dT'}} dT \Rightarrow \frac{L}{A} \approx \frac{1}{I \sqrt{\rho}} \int_{T_L}^{T_H} \frac{\kappa}{\sqrt{Q_H^2 / (I^2 \rho) + 2 \int_T^{T_H} \kappa dT'}} dT$$

Note that the second integral above is an expression relating  $Q_H^2 / (I^2 \rho)$  to the temperatures and  $\kappa$  only, and again this integration can be done analytically.

$$\int_{T_L}^{T_H} \frac{\kappa}{\sqrt{Q_H^2 / (I^2 \rho) + 2 \int_T^{T_H} \kappa dT'}} dT = \sqrt{Q_H^2 / (I^2 \rho) + 2 \int_{T_L}^{T_H} \kappa dT'} - \sqrt{Q_H^2 / (I^2 \rho)}$$

$$\therefore I^2 \rho \frac{L}{A} = \sqrt{Q_H^2 + 2 I^2 \int_{T_L}^{T_H} \rho \kappa dT'} - Q_H \Rightarrow Q_H = \frac{A}{L} \int_{T_L}^{T_H} \kappa dT - \frac{1}{2} I^2 \rho \frac{L}{A}$$

So for nearly constant resistivity  $\rho$ , the heat transfer from the hot plate is just the 0-current thermal heat transfer minus 1/2 of the wire's Ohmic dissipation. The heat transfer to the cold plate must be this thermal heat transfer + 1/2 the Ohmic dissipation (because of energy conservation).

$$\therefore \boxed{Q_L = \frac{A}{L} \int_{T_L}^{T_H} \kappa dT + \frac{1}{2} I^2 \rho \frac{L}{A}}$$

If the current in the wire is greater than the optimal current for the wire's L/A ratio, then the wire will have an area with temperature  $> T_H$ , the temperature of the hot surface. At the point of maximum temperature on the wire, the heat transfer along the wire vanishes, since there is no thermal gradient at that point. Thus between this point and either the end at  $T_L$  or  $T_H$ , the wire has an optimal L/A ratio for the wire current and the temperatures at the ends of that segment. The location of the point of maximum wire temperature and the temperature at that point are (valid for nearly constant resistivity  $\rho$ ):

$$\frac{x}{L} = \frac{\frac{A}{L} \int_{T_L}^{T_H} \kappa dT}{I^2 \rho \frac{L}{A}} + \frac{1}{2} = \frac{\text{Thermal heat transfer}}{\text{Ohmic dissipation}} + \frac{1}{2}$$

$$\int_{T_L}^{T(x)} \kappa dT = \frac{1}{2} I^2 \rho \frac{x^2}{A^2}$$

---

## Calculations of heating for any wire bundle carrying current

- A function to find  $Q_\Omega$ , the 1/2 of the total Ohmic heating, returning a result in  $\mu\text{Watt}$ , given the wire bundle dimensions:

```
In[55]:= Clear[QΩ];

QΩ[ratio_(* L/A *), current_(* mA *)] :=  $\frac{1}{2}$  current^2 rho ratio Meter

Evaluate[Convert[1.0 Micro Ohm Centi Meter (Milli Ampere)^2 / Meter, Micro Watt]]

QΩ[n_(* number of wires *), Len_(* inches *), dia_(* mils *), current_] :=
  QΩ[LbyA[n, Len, dia], current]
```

- A function to find total  $Q_L$ , returning a result in  $\mu\text{Watt}$ , given the wire bundle dimensions:

```
In[58]:= Clear[QL];

QL[ratio_(* L/A *), current_(* mA *)] := Heat[ratio] + QΩ[ratio, current]

QL[n_(* number of wires *), Len_(* inches *), dia_(* mils *), current_] :=
  QL[LbyA[n, Len, dia], current]
```

- A function to find  $T_{\max}$  for a wire connecting 4K and 15K:

```
In[61]:= Clear[Tmax];

Tmax[ratio_(* L/A *), current_ /; current ≤ 0(* mA *)] := 15.0
Tmax[ratio_(* L/A *), current_(* mA *)] :=
  15.0 /; ratio Meter < LbyAopt[1] Meter / current
Tmax[ratio_(* L/A *), current_(* mA *)] := Block[
  {p = 2 QΩ[ratio, current], x, t},
  x = .5 + Heat[ratio] / p;
  t /. FindRoot[10^6 TI[4.0, t] / (ratio Meter) == 1 / 2 p / (Micro Watt) x^2, {t, 15.0}]
]

Tmax[n_(* number of wires *), Len_(* inches *), dia_(* mils *), current_] :=
  Tmax[LbyA[n, Len, dia], current]
```

A function to find  $T_{\max}$  for a wire connecting two specified temperatures:

```
In[66]:= Tmax["T", ratio_(* L/A *), current_ /; current ≤ 0
  (* mA *), Tlow_(* K *), Thigh_(* K *)] := Thigh
Tmax["T", ratio_(* L/A *), current_(* mA *), Tlow_(* K *), Thigh_(* K *)] :=
  Thigh /; ratio Meter < LbyAopt[1, Tlow, Thigh] Meter / current
Tmax["T", ratio_(* L/A *), current_(* mA *), Tlow_(* K *), Thigh_(* K *)] := Block[
  {p = 2 QΩ[ratio, current], x, t},
  x = .5 + Heat["T", ratio, Tlow, Thigh] / p;
  t /. FindRoot[106 TI[Tlow, t] / (ratio Meter) = 1 / 2 p / (Micro Watt) x2, {t, Thigh}]
]
```

---

## Optimal average heat transfer for an intermittent current

If a wire bundle carries a current only intermittently, then the optimum  $L/A$  to minimize the average heat transfer to the cold stage will be larger than the  $L/A$  for a wire carrying the same current continuously. If the fraction of time the current is on is given by  $f$ ,  $0 < f \leq 1$ , then the average heat flow to the cold surface will be:

$$Q_L = \frac{A}{L} \int_{T_L}^{T_H} \kappa dT + f \times \left( \frac{1}{2} I^2 \rho \right) \frac{L}{A} = \frac{A}{L} \int_{T_L}^{T_H} \kappa dT + \frac{1}{2} (I \sqrt{f})^2 \rho \frac{L}{A}$$

So the average heat flow is equivalent to that for a wire with the continuous current scaled by  $\sqrt{f}$ , which is just the RMS current, as expected. Thus the optimal  $A/L$  and the resultant heat flow both scale as  $\sqrt{f}$ :

$$\begin{aligned} Q_L &= I \sqrt{2 f \int_{T_L}^{T_H} \rho \kappa dT} = (Q_L)_{f=1} \sqrt{f} \\ \frac{L}{A} &= \frac{1}{I \sqrt{f \rho}} \sqrt{2 \int_{T_L}^{T_H} \kappa dT} = \left( \frac{L}{A} \right)_{f=1} \frac{1}{\sqrt{f}} \end{aligned}$$

## Choosing among materials

In a practical situation involving cryogenic wire selection, there is probably some size or wire strength constraint on the maximum L/A that can be used. Given the RMS current the wire is expected to carry, the ratio of Ohmic heating to 0-current heat transfer for this maximal L/A is:

$$\frac{Q_{\Omega}}{Q_0} = \frac{f I^2 \hat{\rho}}{2 \hat{\kappa} \Delta T} \left( \frac{L}{A} \right)^2$$

where  $\hat{\rho}$  and  $\hat{\kappa}$  are estimates of their average values over the temperature range  $T_L$  to  $T_H$ , and  $\Delta T = T_H - T_L$ . If  $Q_{\Omega} \ll Q_0$  for the maximum useable L/A for all wire materials being considered, then the presence of the current is negligible, and the wire with the smallest thermal integral over the temperature difference should be chosen.

If, on the other hand,  $Q_{\Omega} \gtrsim Q_0$  for the maximum L/A, then the current flow is not negligible, and the material with the smallest average  $\rho \kappa$  product over the temperature range  $T_L$  to  $T_H$  should be chosen. If  $Q_{\Omega} \gg Q_0$  unless L/A is made too small to be practical, then consider materials with low  $\hat{\rho}$  (even good conductors such as copper or aluminum may be the best choice in some cases). Of course, if  $T_H$  is low enough, superconducting wire would be a good choice.

### ■ Routines to build and format the wire tables:

```
In[69]:= Tlow = 4.0; Thigh = 15.0;
SetMaterial["Phosphor Bronze"];

round[v_, n_] := 10.^-n Round[10.^n v]
round[v_] := round[v, 2]
entry[c_, f_] :=
Block[{ratio = LbyAopt[c Sqrt[f], Tlow, Thigh]},
Style[Grid[{
{round[LbyA[1., 5.] / ratio, 3], "wires/in",
round[OptHeat[c Sqrt[f], Tlow, Thigh] / (Micro Watt)], "μW avg"},
{round[Heat["T", ratio, Tlow, Thigh] / (Micro Watt)], "μW(off)",
round[
Heat["T", ratio, Tlow, Thigh] / (Micro Watt) + QΩ[ratio, c] / (Micro Watt)], "μW(on)"},
{round[Tmax["T", ratio, c, Tlow, Thigh], 1], "K max wire T", SpanFromLeft}
}], Alignment → {{Right, Left, Right, Left}, Automatic},
ItemStyle → {Automatic, {Bold, Automatic, Automatic}},
Background → {None, {Lighter[Yellow, .7], Lighter[Green, .88], Lighter[Green, .88]}},
Dividers → {False, {2 → Gray}}, Spacings → {.2, Automatic}], Small]
]
```

```

In[74]:= currents = {1, 5, 10, 20};
fractions = Range[0.1, 1, .1];

MakeTable :=
(Text@Labeled[
  Grid[
    Transpose[Prepend[Transpose[
      Prepend[
        Table[entry[c, f], {f, fractions}, {c, currents}],
        Style[#, Bold, Larger] & /@ currents
      ]], Style[#, Bold, Larger] & /@
      Join[{Column[{"Current (mA)", "On fraction "}, Alignment → {{Right, Left},
        Automatic}, ItemStyle → Directive[Small, Italic]]}, fractions]]],
    Frame → All, ItemSize → Full],
  Column[{Style["Optimal Bundles of 5mil " <> material <> " Wire", Larger, Bold, Italic],
    Style[Row[{"Temperature: ", Thigh, "K - ", Tlow, "K"}], Larger, Bold, Italic]},
    Center],
  Top, Spacings → {Automatic, 2}
])

```



## Phosphor bronze magnet wiring for CSO dewars

```
In[77]:= currents = {5, 10, 15, 20};
fractions = Range[0.1, 1, .1];
Tlow = 4; Thigh = 15;
SetMaterial["Phosphor Bronze"];
MakeTable
```



### Optimal Bundles of 5mil Phosphor Bronze Wire Temperature: 15K – 4K

| Current (mA)<br>On fraction | 5   | 10  | 15  | 20  |
|-----------------------------|---|---|---|---|
| <b>0.1</b>                  | <b>0.096 wires/in 4.61 <math>\mu</math>W avg</b>  | <b>0.191 wires/in 9.22 <math>\mu</math>W avg</b>  | <b>0.287 wires/in 13.84 <math>\mu</math>W avg</b> | <b>0.383 wires/in 18.45 <math>\mu</math>W avg</b> |
|                             | 2.31 $\mu$ W(off) 25.37 $\mu$ W(on)               | 4.61 $\mu$ W(off) 50.74 $\mu$ W(on)               | 6.92 $\mu$ W(off) 76.1 $\mu$ W(on)                | 9.22 $\mu$ W(off) 101.47 $\mu$ W(on)              |
|                             | 25.1 K max wire T                                 | 25.1 K max wire T                                 | 25.1 K max wire T                                 | 25.1 K max wire T                                 |
| <b>0.2</b>                  | <b>0.135 wires/in 6.52 <math>\mu</math>W avg</b>  | <b>0.271 wires/in 13.05 <math>\mu</math>W avg</b> | <b>0.406 wires/in 19.57 <math>\mu</math>W avg</b> | <b>0.541 wires/in 26.09 <math>\mu</math>W avg</b> |
|                             | 3.26 $\mu$ W(off) 19.57 $\mu$ W(on)               | 6.52 $\mu$ W(off) 39.14 $\mu$ W(on)               | 9.78 $\mu$ W(off) 58.71 $\mu$ W(on)               | 13.05 $\mu$ W(off) 78.27 $\mu$ W(on)              |
|                             | 19.5 K max wire T                                 | 19.5 K max wire T                                 | 19.5 K max wire T                                 | 19.5 K max wire T                                 |
| <b>0.3</b>                  | <b>0.166 wires/in 7.99 <math>\mu</math>W avg</b>  | <b>0.331 wires/in 15.98 <math>\mu</math>W avg</b> | <b>0.497 wires/in 23.97 <math>\mu</math>W avg</b> | <b>0.663 wires/in 31.96 <math>\mu</math>W avg</b> |
|                             | 3.99 $\mu$ W(off) 17.31 $\mu$ W(on)               | 7.99 $\mu$ W(off) 34.62 $\mu$ W(on)               | 11.98 $\mu$ W(off) 51.93 $\mu$ W(on)              | 15.98 $\mu$ W(off) 69.24 $\mu$ W(on)              |
|                             | 17.5 K max wire T                                 | 17.5 K max wire T                                 | 17.5 K max wire T                                 | 17.5 K max wire T                                 |
| <b>0.4</b>                  | <b>0.191 wires/in 9.22 <math>\mu</math>W avg</b>  | <b>0.383 wires/in 18.45 <math>\mu</math>W avg</b> | <b>0.574 wires/in 27.67 <math>\mu</math>W avg</b> | <b>0.765 wires/in 36.9 <math>\mu</math>W avg</b>  |
|                             | 4.61 $\mu$ W(off) 16.14 $\mu$ W(on)               | 9.22 $\mu$ W(off) 32.29 $\mu$ W(on)               | 13.84 $\mu$ W(off) 48.43 $\mu$ W(on)              | 18.45 $\mu$ W(off) 64.57 $\mu$ W(on)              |
|                             | 16.4 K max wire T                                 | 16.4 K max wire T                                 | 16.4 K max wire T                                 | 16.4 K max wire T                                 |
| <b>0.5</b>                  | <b>0.214 wires/in 10.31 <math>\mu</math>W avg</b> | <b>0.428 wires/in 20.63 <math>\mu</math>W avg</b> | <b>0.642 wires/in 30.94 <math>\mu</math>W avg</b> | <b>0.855 wires/in 41.25 <math>\mu</math>W avg</b> |
|                             | 5.16 $\mu$ W(off) 15.47 $\mu$ W(on)               | 10.31 $\mu$ W(off) 30.94 $\mu$ W(on)              | 15.47 $\mu$ W(off) 46.41 $\mu$ W(on)              | 20.63 $\mu$ W(off) 61.88 $\mu$ W(on)              |
|                             | 15.8 K max wire T                                 | 15.8 K max wire T                                 | 15.8 K max wire T                                 | 15.8 K max wire T                                 |
| <b>0.6</b>                  | <b>0.234 wires/in 11.3 <math>\mu</math>W avg</b>  | <b>0.469 wires/in 22.6 <math>\mu</math>W avg</b>  | <b>0.703 wires/in 33.89 <math>\mu</math>W avg</b> | <b>0.937 wires/in 45.19 <math>\mu</math>W avg</b> |
|                             | 5.65 $\mu$ W(off) 15.06 $\mu$ W(on)               | 11.3 $\mu$ W(off) 30.13 $\mu$ W(on)               | 16.95 $\mu$ W(off) 45.19 $\mu$ W(on)              | 22.6 $\mu$ W(off) 60.26 $\mu$ W(on)               |
|                             | 15.4 K max wire T                                 | 15.4 K max wire T                                 | 15.4 K max wire T                                 | 15.4 K max wire T                                 |
| <b>0.7</b>                  | <b>0.253 wires/in 12.2 <math>\mu</math>W avg</b>  | <b>0.506 wires/in 24.41 <math>\mu</math>W avg</b> | <b>0.759 wires/in 36.61 <math>\mu</math>W avg</b> | <b>1.012 wires/in 48.81 <math>\mu</math>W avg</b> |
|                             | 6.1 $\mu$ W(off) 14.82 $\mu$ W(on)                | 12.2 $\mu$ W(off) 29.64 $\mu$ W(on)               | 18.3 $\mu$ W(off) 44.45 $\mu$ W(on)               | 24.41 $\mu$ W(off) 59.27 $\mu$ W(on)              |
|                             | 15.2 K max wire T                                 | 15.2 K max wire T                                 | 15.2 K max wire T                                 | 15.2 K max wire T                                 |
| <b>0.8</b>                  | <b>0.271 wires/in 13.05 <math>\mu</math>W avg</b> | <b>0.541 wires/in 26.09 <math>\mu</math>W avg</b> | <b>0.812 wires/in 39.14 <math>\mu</math>W avg</b> | <b>1.082 wires/in 52.18 <math>\mu</math>W avg</b> |
|                             | 6.52 $\mu$ W(off) 14.68 $\mu$ W(on)               | 13.05 $\mu$ W(off) 29.35 $\mu$ W(on)              | 19.57 $\mu$ W(off) 44.03 $\mu$ W(on)              | 26.09 $\mu$ W(off) 58.71 $\mu$ W(on)              |
|                             | 15.1 K max wire T                                 | 15.1 K max wire T                                 | 15.1 K max wire T                                 | 15.1 K max wire T                                 |
| <b>0.9</b>                  | <b>0.287 wires/in 13.84 <math>\mu</math>W avg</b> | <b>0.574 wires/in 27.67 <math>\mu</math>W avg</b> | <b>0.861 wires/in 41.51 <math>\mu</math>W avg</b> | <b>1.148 wires/in 55.35 <math>\mu</math>W avg</b> |
|                             | 6.92 $\mu$ W(off) 14.61 $\mu$ W(on)               | 13.84 $\mu$ W(off) 29.21 $\mu$ W(on)              | 20.76 $\mu$ W(off) 43.82 $\mu$ W(on)              | 27.67 $\mu$ W(off) 58.42 $\mu$ W(on)              |
|                             | 15. K max wire T                                  | 15. K max wire T                                  | 15. K max wire T                                  | 15. K max wire T                                  |
| <b>1.</b>                   | <b>0.302 wires/in 14.59 <math>\mu</math>W avg</b> | <b>0.605 wires/in 29.17 <math>\mu</math>W avg</b> | <b>0.907 wires/in 43.76 <math>\mu</math>W avg</b> | <b>1.21 wires/in 58.34 <math>\mu</math>W avg</b>  |
|                             | 7.29 $\mu$ W(off) 14.59 $\mu$ W(on)               | 14.59 $\mu$ W(off) 29.17 $\mu$ W(on)              | 21.88 $\mu$ W(off) 43.76 $\mu$ W(on)              | 29.17 $\mu$ W(off) 58.34 $\mu$ W(on)              |
|                             | 15. K max wire T                                  | 15. K max wire T                                  | 15. K max wire T                                  | 15. K max wire T                                  |

## Manganin magnet wiring for CSO dewars

```
In[82]:= currents = {5, 10, 15, 20};
fractions = Range[0.1, 1, .1];
Tlow = 4; Thigh = 15;
SetMaterial["Manganin"];
MakeTable
```

### Optimal Bundles of 5mil Manganin Wire Temperature: 15K – 4K

| Current (mA)<br>On fraction | 5   | 10  | 15  | 20  |
|-----------------------------|---|---|---|---|
| <b>0.1</b>                  | <b>0.336 wires/in 6.41 <math>\mu</math>W avg</b>  | <b>0.672 wires/in 12.82 <math>\mu</math>W avg</b> | <b>1.008 wires/in 19.23 <math>\mu</math>W avg</b> | <b>1.344 wires/in 25.64 <math>\mu</math>W avg</b> |
|                             | 3.2 $\mu$ W(off) 35.25 $\mu$ W(on)                | 6.41 $\mu$ W(off) 70.5 $\mu$ W(on)                | 9.61 $\mu$ W(off) 105.76 $\mu$ W(on)              | 12.82 $\mu$ W(off) 141.01 $\mu$ W(on)             |
|                             | 26.3 K max wire T                                 | 26.3 K max wire T                                 | 26.3 K max wire T                                 | 26.3 K max wire T                                 |
| <b>0.2</b>                  | <b>0.475 wires/in 9.06 <math>\mu</math>W avg</b>  | <b>0.95 wires/in 18.13 <math>\mu</math>W avg</b>  | <b>1.426 wires/in 27.19 <math>\mu</math>W avg</b> | <b>1.901 wires/in 36.26 <math>\mu</math>W avg</b> |
|                             | 4.53 $\mu$ W(off) 27.19 $\mu$ W(on)               | 9.06 $\mu$ W(off) 54.39 $\mu$ W(on)               | 13.6 $\mu$ W(off) 81.58 $\mu$ W(on)               | 18.13 $\mu$ W(off) 108.77 $\mu$ W(on)             |
|                             | 20.1 K max wire T                                 | 20.1 K max wire T                                 | 20.1 K max wire T                                 | 20.1 K max wire T                                 |
| <b>0.3</b>                  | <b>0.582 wires/in 11.1 <math>\mu</math>W avg</b>  | <b>1.164 wires/in 22.2 <math>\mu</math>W avg</b>  | <b>1.746 wires/in 33.3 <math>\mu</math>W avg</b>  | <b>2.328 wires/in 44.41 <math>\mu</math>W avg</b> |
|                             | 5.55 $\mu$ W(off) 24.05 $\mu$ W(on)               | 11.1 $\mu$ W(off) 48.11 $\mu$ W(on)               | 16.65 $\mu$ W(off) 72.16 $\mu$ W(on)              | 22.2 $\mu$ W(off) 96.21 $\mu$ W(on)               |
|                             | 17.8 K max wire T                                 | 17.8 K max wire T                                 | 17.8 K max wire T                                 | 17.8 K max wire T                                 |
| <b>0.4</b>                  | <b>0.672 wires/in 12.82 <math>\mu</math>W avg</b> | <b>1.344 wires/in 25.64 <math>\mu</math>W avg</b> | <b>2.016 wires/in 38.46 <math>\mu</math>W avg</b> | <b>2.688 wires/in 51.28 <math>\mu</math>W avg</b> |
|                             | 6.41 $\mu$ W(off) 22.43 $\mu$ W(on)               | 12.82 $\mu$ W(off) 44.87 $\mu$ W(on)              | 19.23 $\mu$ W(off) 67.3 $\mu$ W(on)               | 25.64 $\mu$ W(off) 89.73 $\mu$ W(on)              |
|                             | 16.6 K max wire T                                 | 16.6 K max wire T                                 | 16.6 K max wire T                                 | 16.6 K max wire T                                 |
| <b>0.5</b>                  | <b>0.751 wires/in 14.33 <math>\mu</math>W avg</b> | <b>1.503 wires/in 28.66 <math>\mu</math>W avg</b> | <b>2.254 wires/in 43. <math>\mu</math>W avg</b>   | <b>3.005 wires/in 57.33 <math>\mu</math>W avg</b> |
|                             | 7.17 $\mu$ W(off) 21.5 $\mu$ W(on)                | 14.33 $\mu$ W(off) 43. $\mu$ W(on)                | 21.5 $\mu$ W(off) 64.49 $\mu$ W(on)               | 28.66 $\mu$ W(off) 85.99 $\mu$ W(on)              |
|                             | 15.9 K max wire T                                 | 15.9 K max wire T                                 | 15.9 K max wire T                                 | 15.9 K max wire T                                 |
| <b>0.6</b>                  | <b>0.823 wires/in 15.7 <math>\mu</math>W avg</b>  | <b>1.646 wires/in 31.4 <math>\mu</math>W avg</b>  | <b>2.469 wires/in 47.1 <math>\mu</math>W avg</b>  | <b>3.292 wires/in 62.8 <math>\mu</math>W avg</b>  |
|                             | 7.85 $\mu$ W(off) 20.93 $\mu$ W(on)               | 15.7 $\mu$ W(off) 41.87 $\mu$ W(on)               | 23.55 $\mu$ W(off) 62.8 $\mu$ W(on)               | 31.4 $\mu$ W(off) 83.73 $\mu$ W(on)               |
|                             | 15.5 K max wire T                                 | 15.5 K max wire T                                 | 15.5 K max wire T                                 | 15.5 K max wire T                                 |
| <b>0.7</b>                  | <b>0.889 wires/in 16.96 <math>\mu</math>W avg</b> | <b>1.778 wires/in 33.92 <math>\mu</math>W avg</b> | <b>2.667 wires/in 50.87 <math>\mu</math>W avg</b> | <b>3.556 wires/in 67.83 <math>\mu</math>W avg</b> |
|                             | 8.48 $\mu$ W(off) 20.59 $\mu$ W(on)               | 16.96 $\mu$ W(off) 41.18 $\mu$ W(on)              | 25.44 $\mu$ W(off) 61.78 $\mu$ W(on)              | 33.92 $\mu$ W(off) 82.37 $\mu$ W(on)              |
|                             | 15.2 K max wire T                                 | 15.2 K max wire T                                 | 15.2 K max wire T                                 | 15.2 K max wire T                                 |
| <b>0.8</b>                  | <b>0.95 wires/in 18.13 <math>\mu</math>W avg</b>  | <b>1.901 wires/in 36.26 <math>\mu</math>W avg</b> | <b>2.851 wires/in 54.39 <math>\mu</math>W avg</b> | <b>3.801 wires/in 72.52 <math>\mu</math>W avg</b> |
|                             | 9.06 $\mu$ W(off) 20.4 $\mu$ W(on)                | 18.13 $\mu$ W(off) 40.79 $\mu$ W(on)              | 27.19 $\mu$ W(off) 61.19 $\mu$ W(on)              | 36.26 $\mu$ W(off) 81.58 $\mu$ W(on)              |
|                             | 15.1 K max wire T                                 | 15.1 K max wire T                                 | 15.1 K max wire T                                 | 15.1 K max wire T                                 |
| <b>0.9</b>                  | <b>1.008 wires/in 19.23 <math>\mu</math>W avg</b> | <b>2.016 wires/in 38.46 <math>\mu</math>W avg</b> | <b>3.024 wires/in 57.69 <math>\mu</math>W avg</b> | <b>4.032 wires/in 76.91 <math>\mu</math>W avg</b> |
|                             | 9.61 $\mu$ W(off) 20.3 $\mu$ W(on)                | 19.23 $\mu$ W(off) 40.59 $\mu$ W(on)              | 28.84 $\mu$ W(off) 60.89 $\mu$ W(on)              | 38.46 $\mu$ W(off) 81.19 $\mu$ W(on)              |
|                             | 15. K max wire T                                  | 15. K max wire T                                  | 15. K max wire T                                  | 15. K max wire T                                  |
| <b>1.</b>                   | <b>1.063 wires/in 20.27 <math>\mu</math>W avg</b> | <b>2.125 wires/in 40.54 <math>\mu</math>W avg</b> | <b>3.188 wires/in 60.81 <math>\mu</math>W avg</b> | <b>4.25 wires/in 81.07 <math>\mu</math>W avg</b>  |
|                             | 10.13 $\mu$ W(off) 20.27 $\mu$ W(on)              | 20.27 $\mu$ W(off) 40.54 $\mu$ W(on)              | 30.4 $\mu$ W(off) 60.81 $\mu$ W(on)               | 40.54 $\mu$ W(off) 81.07 $\mu$ W(on)              |
|                             | 15. K max wire T                                  | 15. K max wire T                                  | 15. K max wire T                                  | 15. K max wire T                                  |